


プチコンでアクションゲームを作る (4)

●コインを出す

ここまでで、自分のキャラクターの動きを作りました。今度は、自分以外のアイテムやキャラクターを画面に出してみましょう。アイテムとしてコインを出して、自分のキャラクターがコインを通過したら、スコアに 100 点が入るようにしてみましょう。



★コインの初期設定

コインは、スプライトの 36 番にあります。4 コマ分の画像があるので、アニメーションで動かしてみましょう。



まず、プログラムの最初に、コインの初期設定をするプログラムを追加します。

1 '*** ACT6 ***	タイトルを「ACT6」に
(※中略)	
10 SPOFS 0, X, Y	
11	
12 'コイン	
13 FOR I=1 TO 10	Iを1～10まで変化させてくりかえし
14 SPSET I, 36, 2, 0, 0, 2	スプライト管理番号1～10 をコインのスプライトにする
15 SPANIM I, 4, 10	4コマ分アニメーション
16 SPOFS I, -50, -50	位置を(-50,-50)=画面外へ
17 SPSETV I, 0, 0	スプライト変数 0 番を、画面に表示されているかどうかの値に 0=画面に出ていない 1=画面に表示されている
18 NEXT I	
19	
20 DY=0	

全体を FOR～NEXT で 10 回繰り返して、コインを 10 個分設定しています。

新しい命令「SPSETV」(エスピーセットブイ)は、スプライトに附属する変数を使って、コインが画面に表示されているか・いないかを区別しています。最初は 0 です。

SPSETV	I,	0,	0
	スプライト	変数番号	変数にセット
	管理番号	(0～7)	する値(数値)

★コインを出すサブルーチン

画面にコインを出すのに、サブルーチンを作って、歩くプログラムから呼び出します。

まずはサブルーチンを呼び出す部分です。

```
72 'チャクチ?
```

(※中略)

```
81 IF C2==0 AND C4!=0 THEN JP=0:DY=0
```

```
82
```

```
83 'コイン
```

```
84 GOSUB @COIN
```

GOSUB でコインのサブルーチンを呼ぶ

```
85
```

```
86 GOTO @ARUKU
```

続いて、コインを出すサブルーチンを、プログラムの最後に追加します。

```
159 ' --- コイン ---
```

```
160 @COIN
```

```
161
```

```
162 CX=RND(239)
```

コインのx座標、y座標を乱数で決定

```
163 CY=RND(177)
```

```
164 CI=RND(10)+1
```

何番目のコインを出すかを乱数で決定

```
165 IF SPGETV(CI,0)==1 THEN RETURN
```

```
166
```

```
167 SPSETV CI,0,1
```

そのコインが既に出ていたら、
コインを出さずに戻る

```
168 SPOFS CI,CX,CY
```

```
169 RETURN
```

コインの0番変数を1にして、
コインの座標を(CX,CY)にして表示

コインをランダムな位置に出したいので、新しく「RND」(ランダム)関数を使っています
RND 関数は、実行する度に違う数字がランダムに出てきます。


```
RND( 10)
```

最大値

※0～(最大値-1)までの乱数が出てくる。この例なら0～9の10種類のどれか。

今回はコインの спраイト管理番号が1～10番なので、「RND(10)+1」として、1～10までの乱数を出しています。

これでプログラムを実行してみましょう。
画面にコインがランダムに出現します。



このままだと、コインがどんどん出現して、あっという間に 10 個になってしまいます。
乱数を使って、少しずつ出現するようにしてみましょう。

```

159 ' --- コイン ---
160 @COIN
161
162 CX=RND(239)
163 CY=RND(177)
164 CI=RND(10)+1
165 IF SPGETV(CI,0)==1 THEN RETURN
166 IF RND(100)>0 THEN RETURN
167
168 SPSETV CI,0,1
169 SPOFS CI,CX,CY
170 RETURN

```

乱数(0~99)が0より大きい時は
コインを出さないで戻る

新しく、乱数でコインを出すか出さないか判断する行を追加しています。
RND(100)で0~99の乱数が発生するので、それが0より大きければリターンしています。
つまり100分の1の確率で、コインが出ることになります。

プログラムを実行してみましょう。
時間が経つに連れて、だんだんコインが表示されます。

●コインをゲットする

表示されたコインを、自分のキャラクターが通過したら 100 点が入るようにしてみましょう。
まず、点数用の変数を作って、画面に点数を表示します。

```

12 'コイン
13 FOR I=1 TO 10
14 SPSET I, 36, 2, 0, 0, 2
15 SPANIM I, 4, 10
16 SPOFS I, -50, -50
17 SPSETV I, 0, 0
18 NEXT I
19
20 DY=0
21 V0=-8
22 G=1
23 JP=0
24 SC=0

```

変数 SC を0にする

点数用の変数を「SC」にして、最初は 0 にしています。

次に、画面に点数を表示しましょう。

画面にはしごを表示した後、左上に点数を表示します。

```

45 'ハシゴ 2F-3F
46 FOR BY=7 TO 14
47 BGPOT 1, 15, BY, 28, 1, 0, 0
48 BGPOT 1, 16, BY, 29, 1, 0, 0
49 NEXT BY
50
51 LOCATE 0, 0
52 PRINT "SCORE "; SC
53
54 ' --- フルク ---

```

カーソルを(0,0)へ移動

点数を表示

新しく「LOCATE」命令が出てきました。これは文字を表示する位置(カーソルの位置)を指定する命令です。

```
LOCATE 0, 0
      x座標 y座標
```

これで、カーソルが画面左上(0,0)に移動して、次の PRINT 命令で文字が表示されます。

そして、プログラムの最後に、コインをゲットするサブルーチンを作ります。

```
181 ' --- コイン ゲット ---
182 @COINGET
183
184 SC=SC+100
185 LOCATE 0,0
186 PRINT "SCORE ";SC
187
188 SPSETV CI,0,0
189 SPOFS CI,-50,-50
190 RETURN
```

点数 SC を 100 増やす

画面左上に点数を表示

コインスプライトの変数0番を0に(出ていない)

コインの座標を画面の外へ

コインに当たったら、点数 SC を100増やして、画面に表示します。

また、このサブルーチンに来た時には、変数 CI が当たったコインの番号になっていますから、SPSETV 命令で CI 番目のコインの変数0番(表示されているかいないか)を0にして、表示されていない状態にします。

また SPOFS 命令で、スプライトの座標を画面の外にします。

プログラムを実行してみましょう。

自分のキャラクターがコインを通過すると、コインが消えて 100 点が入ります。


これで、コインに関するプログラムができました。

ファイル名「ACT6」で保存(SAVE)してください。

SAVE "ACT6"

● 敵を出す

いよいよ敵のキャラクターを出してみましょう。
敵が出ると、ゲームがおもしろくなります。



★ 敵の初期設定

敵のキャラクターは、右向きに歩く敵と左向きに歩く敵の2種類を
考えます。それぞれスプライトの 128 番と 132 番です。



2 コマ分の画像があるので、アニメーションで歩かせてみましょう。

まず、プログラムの最初に、敵の初期設定をするプログラムを追加します。

1	'*** ACT7 ***	タイトルを「ACT7」に
2		
3	ACLS	
4	CLEAR	プログラム実行時に、必ず 変数をクリアする
5		
6	'サイショ (※中略)	
18	SPSETV I, 0, 0	
19	NEXT I	
20		敵の座標を配列変数で指定
21	'うキ	スプライト管理番号 21 番を敵 のスプライトにする
22	DIM TX(2), TY(2)	
23	SPSET 21, 128, 2, 0, 0, 2	2コマ分アニメーション
24	SPANIM 21, 2, 10	
25	SPOFS 21, -50, -50	位置を(-50,-50)=画面外へ
26	SPSETV 21, 0, 0	スプライト変数 0 番を、画面に 表示されているかどうかの値に 0=画面に出ていない 1=画面に表示されている
27	SPSET 22, 132, 2, 0, 0, 2	
28	SPANIM 22, 2, 10	
29	SPOFS 22, -50, -50	
30	SPSETV 22, 0, 0	
31		
32	DY=0	あとは、22 番(左向きの敵) について、同様に設定

まず、4行目で「CLEAR」(クリア)命令を入れています。これは変数を全てクリアする命令です。

次に、コインの設定の下に、敵の設定プログラムを追加しています。

コインは一度出したらその場に止まっているのですが、敵は表示した後に動かさないといいません。

2種類の敵の x 座標、y 座標を管理するのに「DIM」(ディメンジョン)命令を使って、**配列**(はいれつ)変数 TX, TY を 2 個ずつ設定しています。

DIM 命令の文法は以下のとおりです。

```
DIM      A(3)
          変数名
          (配列の個数)
```

配列の個数	例えば A(3)と指定すると、A(0)、A(1)、A(2)の 3 個の配列変数 ^そ が使える。 ()内の数字(添え字)は必ず 0 から始まる。 複数の配列変数を使いたい時は、「,」(カンマ)で区切って書く。
-------	--

配列変数とは番号が付いた変数で、今回の「DIM TX(2),TY(2)」では、TX(0)、TX(1)、TY(0)、TY(1)の 4 つの変数を設定しています。

★敵を登場させる

「歩く」プログラムの中で、敵を登場させるサブルーチンを呼びましょう。

```
102 ' コイン アタリ ハンタイ
103 FOR CI=1 TO 10
104 IF SPHITSP(0, CI) THEN GOSUB @COIN
GET
105 NEXT CI
106
107 ' うキ
108 GOSUB @TEKI
109
110 GOTO @ARUKU
```


そして、プログラムの最後に、敵を登場させるサブルーチンを作ります。

```

207 ' --- テキ ---
208 @TEKI
209
210 TI=RND(2)+21
211 IF SPGETV(TI,0)=1 THEN RETURN
212 IF RND(100)>1 THEN RETURN
213
214 I=TI-21
215 IF TI==21 THEN TX(I)=0 ELSE TX
(I)=239
216 TY(I)=RND(3)*64+40
217
218 SPOFS TI, TX(I), TY(I)
219 SPSETV TI, 0, 1
220 RETURN

```

乱数で、登場させる敵を 21 番か 22 番かどちらかにする

既に登場していたら、出さずにリターン


乱数で 50 分の 1 の確率に入らなかったらリターン

登場する x 座標は、右向き敵なら左端、左向き敵なら右端

y 座標を 1~3 階のどれかの床に

敵を登場させる

登場する敵(右向きか左向きか)を乱数で決めて、それぞれの場合に分けて座標を設定しています。



ここまででプログラムを実行すると、画面の端に敵が登場します。
(敵はまだ動きません)

★敵を動かす

敵が登場した後に、床の上を横へ動くようにしましょう。

プログラムを簡単にするために、左に登場した敵は右へ、右に登場した敵は左へまっすぐ移動して、端に着いたら消えるようにします。



「歩く」プログラムの中で、敵を移動させるサブルーチン呼びましょう。

```

107 ' うキ
108 GOSUB @TEKI
109
110 ' うキ イトウ
111 TI=21
112 GOSUB @TEKIMOVE
113 TI=22
114 GOSUB @TEKIMOVE
115
116 GOTO @ARUKU

```

21 番(右向きの敵)を動かすサブルーチンと呼ぶ

22 番(左向きの敵)を動かすサブルーチンと呼ぶ

敵が右向き・左向きの2種類いるので、変数 TI に番号を設定して、サブルーチンを2回呼びます。

そして、プログラムの最後に、敵を動かすサブルーチンを作ります。

```

228 ' --- うキ イトウ ---
229 @TEKIMOVE
230
231 IF SPGETV(TI,0)==0 THEN RETURN
232 I=TI-21
233 IF TI==21 THEN TX(I)=TX(I)+1 E
LSE TX(I)=TX(I)-1
234 IF TX(I)<-15 OR TX(I)>255 THEN
SPSETV TI,0,0
235 SPOFS TI,TX(I),TY(I)
236 RETURN

```

敵が登場していなかったらリターン

敵を左右に動かす

敵が端まで行ったら消す

敵の座標を変更して移動させる

プログラムを実行してみましょう。敵が登場して、左右に移動します。

●敵との当たり判定

自分と敵との当たり判定を作りましょう。

プログラムを簡単にするために、敵に当たったら即座にゲームオーバーになるようにしてみよう(きびしいルールですが)。

「歩く」プログラムで、敵との当たり判定を追加します。

```

110 ' うき イトウ
111 TI=21
112 GOSUB @TEKIMOVE
113 TI=22
114 GOSUB @TEKIMOVE
115
116 ' うき アタリ ハンタイ
117 IF SPHITSP(0, 21) THEN GOTO @TEKI
HIT
118 IF SPHITSP(0, 22) THEN GOTO @TEKI
HIT
119
120 GOTO @ARUKU

```

自分と敵が当たったら、@TEKIHIT へジャンプ

コインの時と同じく SPHITSP 関数で、自分(0番)と敵(21番、22番)のSpriteの当たり判定をします。

当たった場合は、敵と当たった処理をするプログラム@TEKIHIT へジャンプします。

そして、敵と当たった処理をするプログラムを、最後に追加します。

```

242 ' --- うき ヒット ---
243 @TEKIHIT
244
245 LOCATE 7, 12
246 PRINT "XXX GAME OVER XXX"
247 END

```

プログラムを実行してみましょう。

敵に当たると、ゲームオーバーになって、プログラムが終了します。

●BGM を付ける

これまで全く音が無い状態だったので、まずは BGM (音楽)を付けてみましょう。

ゲームがスタートしたら、音楽が流れるようにします。

ゲームが始まる所で、「BGMPLAY」(बीジーエムプレイ)命令を入れます。

```

1 '*** ACT7 ***
2
3 ACLS
4 CLEAR
5 BGMPLAY 14
6 'サイショ

```

プログラムを実行すると、ゲーム開始から曲が流れます。

BGMPLAY 命令の文法は、以下のとおりです。

```

BGMPLAY 23
          曲番号      曲番号は 0～29。

```

曲は 30 曲あります。一覧リストは→を見てください。
曲番号を変えると、違う曲になります。

さらに、ゲームオーバーになった時の BGM も付けてみましょう。

```

243 '--- テキ ヒット ---
244 @TEKIHIT
245
246 BGMPLAY 6
247 LOCATE 7,12
248 PRINT "XXX GAME
OVER XXX"
249 END

```

番号	説明
0	軽快な曲
1	湿った暗い感じの曲
2	緊張感高まる曲
3	激しくアップテンポな曲
4	スタートジングル
5	クリアジングル
6	ゲームオーバー
7	メニューセレクト
8	結果発表
9	スタッフロール
10	スタッフロール その2
11	時代劇ゲーム風
12	軽快なマーチバンド風
13	激しいロック調
14	軽快な曲 その2
15	WOND
16	考え中
17	WOND2
18	未来系
19	BAL
20	BAL_2
21	スパイ系
22	SCI
23	シューティング
24	パッド
25	SEN
26	ピュア
27	ROA
28	CUR
29	FIG

●効果音を付ける

自分のキャラクターがジャンプする時、コインを取った時などの効果音を付けてみましょう。
効果音を出すには、「BEEP」(ビーブ)命令を使います。
文法は以下のとおりです。

BEEP 0 , 0 , 127 , 64
 波形 ピッチ 音量 パン
 番号 ポット

波形番号	0～69。詳しくは下の表を見てください。省略時は0。
ピッチ	音の高さ。0=標準。-8192=2 オクターブ下、8192=2 オクターブ上。
音量	0(無音)～127(最大)。
パンポット	ステレオスピーカーを使って、音を左右にふる。 0=左、64=中央、127=右。

※それぞれの値は省略可能。

波形番号(0～69)は以下のとおりです。いろいろな効果音の他、楽器の音もあります。

0	BEEP	18	シンセベース	36	AUTO	54	ダンス HH
1	ノイズ	19	シンセベース	37	キラ	55	ヒット
2	カーソル移動	20	ギター	38	ESC	56	ティンパレス
3	決定	21	オルガン	39	バンジョー2	57	チャイニーズシンバル
4	キャンセル	22	ピアノ	40	スクラッチ	58	ミニシンバル
5	上昇	23	カウベル	41	ギター2	59	シェーカー
6	下降	24	タム	42	オルガン2	60	鈴
7	コイン	25	シンバル	43	ピアノ2	61	太鼓
8	ジャンプ	26	オープンハイハット	44	PASS	62	シンセ
9	着地	27	クローズハイハット	45	UP2	63	かっこう
10	発射	28	ハンドクラップ	46	録音	64	パフ!
11	ダメージ	29	リムショット	47	シンセタム	65	nohkan
12	金属	30	スネアドラム	48	カウベル2	66	humandr1
13	爆発	31	バスドラム	49	metro	67	humandr2
14	叫び声	32	OK2	50	tri	68	犬
15	ブレーキ	33	BALL	51	コンガ	69	猫
16	バンジョー	34	和風	52	ダンス BD		
17	シンセストリングス	35	VOLT	53	ダンス SD		

ジャンプの効果音は、ジャンプし始める時に鳴らしましょう。

```
172 ' --- ジャンプ ---  
(中略)  
189 @JUMP2  
190 BEEP 8  
191 JP=1  
192 DY=V0  
193 RETURN
```

コインを取った時の効果音は、「コイン ゲット」の所で鳴らしましょう。

```
208 ' --- コイン ゲット ---  
209 @COINGET  
210  
211 BEEP 7  
212 SC=SC+100  
213 LOCATE 0,0  
214 PRINT "SCORE ";SC
```

プログラムを実行してみましょう。

ゲームらしく、かなりにぎやかになります。